MeanRev-EuroFX_30m – PowerLanguage code
//EuroFX 30min timeframe
Inputs:
    Begin_Time(0230),//2.30amChicago = 8.30amLondon
    End_Time(1430),//14.30pmChicago = 20.30pmLondon
    Len(18),
    Len2(28),
    OB(71), 
    OS(20),  
    BB_SD(1.5),
    emaLenLX(27),
    emaLenSX(28), 
    adxLen(5), 
    minchng(1), 
    TakeProfit_ticks(260),
    Stoploss_ticks(140),
    MySize(1);
  
Variables: 
    UpperBB(0), 
    LowerBB(0),
    emaLX(0),
    emaSX(0),   
    adxval(0),
    rsimfi(0),
    BuySetup(false),
    SellSetup(false),
    mp(0);
    
// Declarations-------------------------------------
UpperBB = BollingerBand(Close,Len,+BB_SD); 
LowerBB = BollingerBand(Close,Len,-BB_SD);
emaLX = XAverage(Close,emaLenLX); 
emaSX = XAverage(Close,emaLenSX); 
rsimfi = (RSIClassic(C,Len2) + MoneyFlow(Len2)) / 2;
adxval = ADX(adxLen);
mp = MarketPosition;

// Setup Conditions--------------------------------
BuySetup = Close[1] < LowerBB[1] and rsimfi < OS;
SellSetup = Close[1] > UpperBB[1] and rsimfi > OB;
       
// Long Entry----------------------------
If time >= Begin_Time and time < End_Time and mp <> 1 and Buysetup then begin
	if Close > LowerBB and (adxval[1] - adxval) > minchng then
		Buy ("LE") MySize contracts next bar market;
end;	
	
// Short Entry----------------------------
If time >= Begin_Time and time <= End_Time and mp <> -1 and Sellsetup then begin
	if Close < UpperBB and (adxval[1] - adxval) > minchng then
		SellShort ("SE") MySize contracts next bar market;
end; 
	
// Exit--------------------------------------
if mp = 1 and Close > emaLX then sell ("LX") Next Bar market;
if mp = -1 and Close < emaSX then buytocover ("SX") Next Bar market;

// Order Management-------------------------
SetStopContract;
if TakeProfit_ticks > 0 then SetProfitTarget(TakeProfit_ticks * TickSize * BigPointValue);
if StopLoss_ticks > 0 then SetStopLoss(Stoploss_ticks * TickSize * BigPointValue);


